Глава 2 Формирование инвестиционного портфеля и прогнозирование рисков

В предыдущей главе частично затронут вопрос о неоднозначности прогнозов относительно курсов акций. В частности, сравнивалась ожидаемая ставка доходности по акциям с аналогичным показателем для облигаций. Решение при этом принималось на основе простого сравнения численных значений. Однако в реальности не все так просто. Дело в том, что ожидаемая ставка доходности — это некое характерное значение, которое может быть нереализуемым в принципе. Например, если ставка процента может принимать только два значения, 4 и 6 %, с равными вероятностями, то ожидаемым будет значение 5 %, и оно в указанных обстоятельствах не будет реализовано никогда. Просто если бы соответствующий инвестиционный проект реализовывался достаточно большое количество раз, в среднем ставка процента была бы близка к 5 %. Это очень простая иллюстрация, и разница между реальной и ожидаемой ставкой процента незначительна. Обычно разброс возможных значений ставки процента существенно шире, а среди возможных есть и такие варианты, когда инвестор не только не получает прибыли, но и несет убытки. Поэтому, какими бы аккуратными и хитроумными ни были предварительные расчеты, инвестирование связано с некоторым риском. Другое дело, что степень этого риска для разных проектов различна. Риск и доходность идут рука об руку — чем выше ожидаемая доходность, тем больше риск ее не получить. При принятии инвестиционного решения приходится искать баланс между приемлемой степенью риска и размерами прибыли. Далее рассмотрим пример, в котором анализируются два инвестиционных проекта, связанных с приобретением различных акций.

Оценка риска при инвестировании

На рис. 2.1 показан документ с данными по акциям трех типов. Для каждой из акций приведен набор возможных значений ставки доходности и вероятности ее реализации. По этим данным вычисляются математическое ожидание для ставки доходности и стандартное отклонение (корень квадратный из дисперсии).

B8 | {=КОРЕНЬ(СУММПРОИЗВ((B5:H5-B7)^2;B6:H6))}

	A	B	C	D	E	F	G	H	I
1				***Анализ рисков***					
2									
3	***Акции А***								
4	Вариант	1	2	3	4	5	6	7	Итого
5	Ставка доходности	8%	6%	5%	3%	1%	-2%	-4%	2%
6	Вероятность	0,07	0,08	0,1	0,4	0,2	0,1	0,05	1
7	Ожидаемая ставка	2,54%							
8	Отклонение	2,90%							
9									
10	***Акции Б***								
11	Вариант	1	2	3	4	5	6	Итого	
12	Ставка доходности	12%	9%	3%	1%	-1%	-8%	3%	
13	Вероятность	0,3	0,15	0,05	0,05	0,15	0,3	1	
14	Ожидаемая ставка	2,60%							
15	Отклонение	8,27%							
16									
17	***Акции В***								
18	Вариант	1	2	3	4	Итого			
19	Ставка доходности	9%	7%	-5%	-3%	2%			
20	Вероятность	0,4	0,1	0,1	0,4	1			
21	Ожидаемая ставка	2,60%							
22	Отклонение	6,05%							
23									

Рис. 2.1. Анализ рисков

Так, например, в ячейку B7 вводится формула =СУММПРОИЗВ(B5:H5;B6:H6) для вычисления ожидаемой ставки доходности (в ячейках B5:H5 представлен набор возможных значений ставки доходности, а в ячейках B6:H6 — вероятность их реализации). Как известно, если случайная величина ξ может принимать значения ξ_k с вероятностями $P(\xi = \xi_k)$, $(k = 1,2,\dots,N)$, то математическое ожидание $M\xi$ случайной величины ξ дается суммой: $M\xi = \sum_{k=1}^{N} \xi_k P(\xi = \xi_k)$. Именно такая сумма реализуется указанной выше формулой. Поскольку стандартное отклонение вычисляется как корень квадратный из дисперсии, необходимо вычислить и ее. Дисперсия $D\xi$ определяется как математическое ожидание квадратов отклонений случайной величины от ее математического ожидания, то есть $D\xi = M(\xi - M\xi)^2$. В случае дискретно распределенной случайной величины имеем $D\xi = \sum_{k=1}^{N} (\xi_k - M\xi)^2 P(\xi = \xi_k)$. Это выражение для рассматриваемого случая в Excel может быть реализовано с помощью формулы =КОРЕНЬ(СУММПРОИЗВ((B5:H5-B7)^2;B6:H6)), которая вводится в ячейку B8 как формула массива (комбинация клавиш Ctrl+Shift+Enter). Это нужно для того, чтобы командой (B5:H5-B7)^2 формировался массив данных, получающихся возведением в квадрат отклонений значений ячеек диапазона B5:H5 от математического ожидания (ячейка B7).

Значение в ячейке I5 дает представление о среднеарифметической ставке доходности (формула =СРЗНАЧ(B5:H5)), а формула =СУММ(B6:H6) в ячейке I6 позволяет контролировать корректность введенных данных для закона распределения случайной величины — сумма всех вероятностей должна равняться единице.

Для двух оставшихся типов акций характеристики распределения вычисляются абсолютно аналогично. Для удобства основные формулы с указанием ячеек, в которые они вводятся, приведены в табл. 2.1.

Таблица 2.1. Значения ячеек таблицы

Ячейка или диапазон	Формула или значение
B12:G12	Набор возможных значений
B13:G13	Вероятности реализации значений
H12	=СРЗНАЧ(B12:G12)
H13	=СУММ(B13:G13)
B14	=СУММПРОИЗВ(B12:G12;B13:G13)
B15	=КОРЕНЬ(СУММПРОИЗВ((B12:G12-B14)^2;B13:G13)) (формула массива)
B19:E19	Набор возможных значений
B20:E20	Вероятности реализации значений
F19	=СРЗНАЧ(B19:E19)
F20	=СУММ(B20:E20)
B21	=СУММПРОИЗВ(B19:E19;B20:E20)
B22	=КОРЕНЬ(СУММПРОИЗВ((B19:E19-B21)^2;B20:E20)) (формула массива)

Вывод относительно выбора нужного типа акций предстоит сделать по значениям математических ожиданий и стандартных отклонений. Общий качественный критерий состоит в том, что математическое ожидание должно быть по возможности больше, а стандартное отклонение как можно меньше. Отсюда можно выработать два практических правила.

1. Если для двух проектов математические ожидания ставок доходности одинаковы, то выбрать следует тот, для которого меньше стандартное отклонение (дисперсия) ожидаемой ставки доходности.

2. Если стандартные отклонения (дисперсии) ожидаемых ставок доходности для двух проектов одинаковы, то выбрать следует проект с большим математическим ожиданием ставки доходности.

К сожалению, обычно приходится выбирать между проектами с абсолютно разными характеристиками и сформулированными правилами воспользоваться удается редко. Многое в этом случае зависит от личных предпочтений инвестора и его склонности к риску. Рассматриваемый случай достаточно простой и можно провести некоторый анализ. Так, для акций второго и третьего типа (см. рис. 2.1) математические ожидания ставок доходности одинаковы, но в третьем случае стандартное отклонение меньше. Это означает, что третий вариант предпочтительнее второго. Теперь сравним третий и первый варианты. Математическое ожидание ставки доходности для акций первого типа несколько (но не очень сильно) меньше математического ожидания ставки доходности для акций третьего типа. Однако стандартное отклонение в первом случае почти вдвое меньше стандартного отклонения для третьего варианта. Кроме того, это стандартное отклонение в случае акций первого типа практически не превышает математическое ожидание, что означает достаточно достоверную неубыточность инвестиций в соответствующий проект. Этого нельзя сказать о третьем проекте. Для него стандартное отклонение более чем вдвое превышает математическое ожидание, то есть возможность оказаться в убытке представляется весьма актуальной. С другой стороны, максимально возможные ставки доходности по обоим проектам практически одинаковы: 8 % для первого проекта и 9 % для третьего, но вероятность получить максимальный процент в третьем проекте почти в шесть раз больше, чем в первом! Из всего сказанного можно сделать заключение, что осторожный инвестор, скорее всего, будет вкладывать средства в акции первого типа, а склонный к риску — не исключено, что в акции третьего типа. Разумеется, можно часть средств вложить в первый проект, а часть — в третий (так обычно и делают). В этом случае речь идет о версификации активов. Эта процедура также может выполняться по-разному. На ней мы еще остановимся.

Выше выбор осуществлялся между проектами с разной степенью рискованности и доходности. В некотором смысле граничной является ситуация, когда есть два проекта: один безрисковый (или почти безрисковый), а другой рискованный, но ожидаемая ставка доходности по нему выше, чем у безрискового проекта. Разницу между ожидаемой ставкой доходности рискованного проекта и ставкой доходности безрискового проекта обычно называют премией за риск. Обозначим через ξ ставку доходности рискованного проекта (это случайная величина), а ставку доходности безрискового проекта будем обозначать как η (это постоянная детерминированная величина). Ожидаемая ставка доходности рискованного проекта тогда равна $M\xi$, а стандартное отклонение $\sigma = \sqrt{D\xi}$. Через x обозначим долю средств, которые инвестируются в рискованный проект (соответственно, в безрисковый проект вкладывается $(1 - x)$ часть средств). Если общая сумма инвестируемых средств равна q, то доход равен $\Delta q = xq\xi + (1 - x)q\eta$, а ставка дохода в этом случае равна $\nu = \Delta q / q = x\xi + (1 - x)\eta$. Поскольку ξ есть случайная величина, то и ν также является величиной случайной. Мы не можем однозначно указать ее будущее значение, но можем определить ожидаемое значение этой случайной величины, то есть определить ее математическое ожидание. Чтобы определить ожидаемую ставку доходности воспользуемся тем, что математическое ожидание суммы равно сумме математических ожиданий, постоянные величины можно выносить за знак математического ожидания, а математическое ожидание детерминированной величины равно самой этой величине. Получим следующее: $M\nu = M[x\xi + (1-x)\eta] = xM\xi + (1-x)\eta$. Чтобы вычислить стандартное отклонение, предварительно найдем дисперсию случайной величины ν. Поскольку для независимых величин, каковыми и являются ξ и η, дисперсия суммы равна сумме дисперсий, постоянный множитель из под знака дисперсии можно выносить с возведением его в квадрат (а дисперсия детерминированной величины равна нулю), то получаем $D\nu = D[x\xi + (1-x)\eta] = D[x\xi] + D[(1-x)\eta] = x^2 D\xi$. Таким образом, стандартное отклонение $\sigma_\nu = \sqrt{D\nu} = x\sigma$. Следовательно, ожидаемая ставка доходности от инвестирования в рассматриваемый проект является линейной функцией параметра x, то есть части средств, инвестированных в рискованные активы. Изменяя x в пределах от 0 до 1, можем изменять ожидаемую ставку доходности от значения η до значения $M\xi$ (в отличие от случайной величины ξ значение $M\xi$ есть величина постоянная!). При этом стандартное отклонение, характеризующее степень возможного отклонения ставки доходности от ожидаемого значения, изменяется от 0 до σ.

Поскольку ожидаемая ставка и стандартное отклонение зависят от параметра x, можем рассматривать параметрическую зависимость ожидаемой ставки от стандартного отклонения. Эта зависимость, очевидно, также является линейной (в силу линейной зависимости ожидаемой ставки и стандартного отклонения от параметра x). Если изобразить зависимость ожидаемой ставки от стандартного отклонения графически, получим линию (точнее, линейный отрезок, поскольку параметр x принимает значения в конечном диапазоне от 0 до 1). Таким образом, выбор точки на соответствующей прямой соответствует выбору определенной инвестиционной стратегии. Критерии выбора точки могут быть самыми разными. В данном случае рассмотрим одну достаточно интересную ситуацию. Будем предполагать, что существует некая, оптимальная с точки зрения инвестора, комбинация значений ожидаемой ставки и стандартного отклонения. При графической интерпретации это точка в координатной плоскости значений отклонения и ожидаемой ставки. Задача сводится к тому, чтобы найти на линии зависимости ожидаемой ставки от стандартного отклонения такую точку, которая наименее удалена от точки оптимальных значений ожидаемой ставки и стандартного отклонения. Здесь в «дебри» графического анализа углубляться не будем, а прибегнем к алгебраической интерпретации. В общем случае задача формулируется как поиск минимума функции $L = k(M\nu - a)^2 + p(\sigma_\nu - b)^2$, где k и p есть феноменологические коэффициенты, а a и b — оптимальные значения ожидаемой ставки и стандартного отклонения соответственно. Задача эта может быть решена аналитически. Для этого достаточно вычислить производную по параметру x и приравнять ее к нулю. В результате получим, что в рискованный проект следует вложить $x = \left(p\sigma b + k(a-\eta)(M\xi-\eta)\right) \Big/ \left(k(M\xi-\eta)^2 + p\sigma^2\right)$ часть средств. Это достаточно громоздкая формула, и она не учитывает одно немаловажное обстоятельство, а именно, то, что параметр x не может превышать 1 (если, конечно, речь не идет о займе средств для инвестирования). Кроме того, когда есть такое эффективное приложение, как Excel, необходимость в аналитических расчетах отпадает. Рассмотрим документ, представленный на рис. 2.2.



B11 =B9*(B12-B6)^2+B10*(B13-B7)^2

	A	B	C	D	E
1	***Распределение рисков***				
2					
3	*Ожидаемая ставка рискованного проекта*	12%			
4	*Стандартное отклонение для рискованного проекта*	13%			
5	*Ставка доходности безрискового проекта*	6%			
6	*Оптимальная ставка доходности*	7%			
7	*Оптимальное стандартное отклонение*	3%			
8	*Часть инвестиций в рискованный проект*	50%	22%		
9	*Коэффициент k*	1			
10	*Коэффициент p*	1			
11	*Минимизируемая функция*	1,63E-03			
12	*Ожидаемая доходность*	9%			
13	*Ожидаемое отклонение*	7%			
14	*Премия за риск*	3%			
15					

Рис. 2.2. Распределение активов между рискованным и безрисковым проектами

Назначение ячеек этого документа с указанием формул, которые в них вводятся, приведено в табл. 2.2.

Таблица 2.2. Назначение ячеек на рис. 2.2

Ячейка	Описание и формула
B3	Ожидаемая ставка рискованного проекта
B4	Стандартное отклонение для рискованного проекта
B5	Ставка доходности безрискового проекта
B6	Оптимальная ставка доходности
B7	Оптимальное стандартное отклонение
B8	Часть инвестиций в рискованный проект

Ячейка	Описание и формула
C8	Проверка оптимального значения параметра, определяющего часть инвестиций в рискованный проект. Вычисляется по формуле =(B10*B4*B7+B9*(B6-B5)*(B3-B5))/(B9*(B3-B5)^2+B10*B4^2)
B9	Коэффициент k
B10	Коэффициент p
B11	Минимизируемая функция. Вычисляется по формуле =B9*(B12-B6)^2+B10*(B13-B7)^2
B12	Ожидаемая доходность. Вычисляется по формуле =B8*B3+(100%-B8)*B5
B13	Ожидаемое отклонение, вычисляемое по формуле =B8*B4
B14	Премия за риск. Дается формулой =B12-B5

В ячейки диапазона B3:B7 вводятся численные значения (в процентном формате), определяющие основные параметры модели, такие как ставка для безрискового проекта, ожидаемая ставка доходности рискованного проекта и т. д. (см. табл. 2.1). Значение в ячейке B8 следует выбрать исходя из условия минимальности определяемой далее функции. Однако, поскольку для этого оптимального значения существует аналитическое выражение, в соседнюю ячейку C8 введена соответствующая формула для вычисления этого значения. После выполнения оптимизации значения в ячейках B8 и C8 должны совпадать.

В ячейки B9 и B10 вводятся численные значения для коэффициентов модели k и p. Наконец, минимизируемая функция вводится в ячейку B11. Формула (см. табл. 2.1) содержит ссылки на ячейки B12 и B13 — значения ожидаемой доходности и ожидаемого отклонения. Эти значения зависят от части инвестированных в рискованный проект средств. Таким образом, минимизируемая функция также зависит от этого параметра. Премия за риск отображается в ячейке B14.

Перед поиском оптимального значения части инвестиций в рискованный проект в ячейку B8 нужно ввести некоторое пробное значении (на рис. 2.2 это значение равно 50%). Решение задачи будем искать с помощью надстройки Excel Поиск решения, которую нужно сначала загрузить (загрузка и работа с данной надстройкой подробно излагаются в разд. «Надстройка Поиск решения» и «Задачи оптимизации» гл. 5). После этого запускаем надстройку Поиск решения. Особенность настройки параметров диалогового окна Поиск решения состоит в том, что в качестве дополнительных условий нужно указать, что значение изменяемой ячейки B8 должно лежать в пределах от 0 до 1. Минимизируем, как несложно догадаться, значение в ячейке B11 (рис. 2.3).

Результат вполне приемлем — во всяком случае, он совпадает с проверочным значением в ячейке C8 (рис. 2.4).



Рис. 2.3. Поиск оптимального значения части инвестиций в рискованный проект



	A	B	C
1	Распределение рисков		
2			
3	Ожидаемая ставка рискованного проекта	12%	
4	Стандартное отклонение для рискованного проекта	13%	
5	Ставка доходности безрискового проекта	6%	
6	Оптимальная ставка доходности	7%	
7	Оптимальное стандартное отклонение	3%	
8	Часть инвестиций в рискованный проект	22%	22%
9	Коэффициент k	1	
10	Коэффициент p	1	
11	Минимизируемая функция	1,22E-05	
12	Ожидаемая доходность	7%	
13	Ожидаемое отклонение	3%	
14	Премия за риск	1%	
15			

Рис. 2.4. Результат оптимизации

Однако, как отмечалось ранее, может возникнуть ситуация, когда оптимальным (если рассчитывать по приведенной выше аналитической формуле) является значение части инвестиций в рискованный проект, большее единицы (или 100 %). Такая ситуация проиллюстрирована рис. 2.5.

C8 f_x =(B10*B4*B7+B9*(B6-B5)*(B3-B5))/(B9*(B3-B5)^2+B10*B4^2)

	A	B	C	D	E	F	G	H
1	***Распределение рисков***							
2								
3	***Ожидаемая ставка рискованного проекта***	10%						
4	***Стандартное отклонение для рискованного проекта***	15%						
5	***Ставка доходности безрискового проекта***	5%						
6	***Оптимальная ставка доходности***	20%						
7	***Оптимальное стандартное отклонение***	15%						
8	***Часть инвестиций в рискованный проект***	100%	**120%**					
9	***Коэффициент k***	1						
10	***Коэффициент p***	1						
11	***Минимизируемая функция***	1,00E-02						
12	***Ожидаемая доходность***	10%						
13	***Ожидаемое отклонение***	15%						
14	***Премия за риск***	5%						
15								

Рис. 2.5. Все средства инвестированы в рискованный проект

Для ее реализации достаточно установить существенно большое значение для оптимальной ставки доходности и стандартного отклонения. В частности, на рис. 2.5 видим, что аналитически рассчитанное значение в ячейке равно 120%, в то время как в результате оптимизации получаем 100% (при этом ожидаемая доходность и стандартное отклонение совпадают с ожидаемой ставкой доходности и стандартным отклонением для рискованного проекта соответственно). Дело в том, что аналитическая формула получена без каких бы то ни было предположений относительно значения части инвестиций в рискованный проект, а оптимизация выполняется с учетом этого ограничения. Поэтому здесь как раз та ситуация, когда Excel проявляет свои преимущества. С другой стороны, значение в ячейке C8 автоматически меняется при изменении исходных данных, а чтобы получить корректное значение в ячейке B8 нужно каждый раз выполнять оптимизацию. Разумный выход может быть в том, чтобы прибегать к оптимизации только в тех случаях, когда значение в ячейке C8 выходит за допустимые пределы.

Пример со значением части инвестиций в рискованный проект, большей 100 %, наводит на некоторые размышления. Действительно, если ожидаемая ставка доходности достаточно высока, можно занять в долг дополнительные средства и вложить их в проект. Деньги (во всяком случае, большие) в долг дают, как правило, под проценты. Причем ставка процента в этом случае превышает ставку дохода по безрисковым активам (иначе тому, кто дает в долг, было бы выгоднее вложить средства в безрисковые активы).

Если средства занимаются под проценты, бóльшие ставки процента по безрисковым активам, эти средства имеет смысл вкладывать только в рискованный проект. Действительно, какой смысл проводить финансовую операцию, потенциальная прибыль от которой не покроет расходов на оплату процентов по займам?

Таким образом, можем рассмотреть более общую, по сравнению с предыдущей, задачу: распределение активов между безрисковым и рискованным проектами с учетом возможности привлечения дополнительных средств. Причем сразу оговоримся, что если в рассмотренной выше модели снять ограничение на значение доли инвестиций в рискованный проект, то она адекватно будет описывать ситуацию, когда заем средств осуществляется под ставку процента, равную ставке доходности безрискового актива. Если это так, можно также использовать аналитическое выражение для доли рискованных инвестиций, причем без всяких ограничений.

В общем случае, когда процент на заем превышает ставку доходности безрискового актива, существует две принципиальные ситуации. Первая имеет место, когда инвестирование осуществляется без привлечения дополнительных средств. Она может быть описана в рамках рассмотренной выше модели. А вот если окажется, что доля рискованных инвестиций должна быть больше 100 %, придется учесть механизм выплат по процентам.

Обозначим ставку заемного процента через α. Тогда ставку доходности можно представить в виде $\nu = x\xi + (1 - x)[\theta(1 - x)\eta + \alpha\theta(x - 1)]$, где $\theta(z)$ есть функция Хэвисайда, равная единице при положительном аргументе и нулю в противном случае. Ожидаемая ставка доходности равна $M\nu = xM\xi + (1 - x)[\theta(1 - x)\eta + \alpha\theta(x - 1)]$, а стандартное отклонение, как и ранее, $\sigma_\nu = x\sigma$. Из всего вышеописанного видим, что, внеся минимальные изменения в исходную модель, можем учесть эффект привлечения заемных средств. На рис. 2.6 показан документ с реализацией этой новой модели.

B13 fx =B9*B3+(100%-B9)*ЕСЛИ(B9<100%;B5;B6)

	A	B	C	D	E	F	G
1	***Распределение рисков***						
2							
3	***Ожидаемая ставка рискованного проекта***	10%					
4	***Стандартное отклонение для рискованного проекта***	15%					
5	***Ставка доходности безрискового проекта***	5%					
6	***Процентная ставка займа***	8%					
7	***Оптимальная ставка доходности***	20%					
8	***Оптимальное стандартное отклонение***	15%					
9	***Часть инвестиций в рискованный проект***	109%					
10	***Коэффициент k***	1					
11	***Коэффициент p***	1					
12	***Минимизируемая функция***	9,83E-03					
13	***Ожидаемая доходность***	10%					
14	***Ожидаемое отклонение***	16%					
15	***Премия за риск***	5%					

Рис. 2.6. Инвестирование с привлечением дополнительных средств

По большому счету, пришлось изменить только одну формулу для ожидаемой ставки доходности (ячейка B13). В эту ячейку вводится формула =B9*B3+(100%-B9)*ЕСЛИ(B9<100%;B5;B6), в которой использована условная функция `ЕСЛИ()`. В зависимости от значения части инвестируемых в рискованный проект средств функцией возвращается либо ставка доходности безрискового проекта, либо ставка по займу. Кроме того, необходимо учесть, что, поскольку в список параметров включена строка для ставки по займу, во всех прочих формулах в соответствии с этим несколько изменились ссылки.

Чтобы получить результат, как на рис. 2.6, нужно выполнить оптимизацию. Входные данные подобраны так, что доля инвестиций в рискованный актив больше 100 %. Однако эта доля меньше того значения, которое было получено в рамках старой модели при расчете по аналитической формуле (см. рис. 2.5). Но так и должно быть, поскольку в данном случае мы учли разницу ставки по займу и ставки доходности безрисковых активов.

Если подобрать другие значения входных данных, можно получить результат, при котором часть средств, вложенных в рискованный проект, меньше 100 %. Желающие могут сравнить документ на рис. 2.7 с документом на рис. 2.4.



B12 =B10*(B13-B7)^2+B11*(B14-B8)^2

	A	B	C	D	E	F
1	***Распределение рисков***					
2						
3	*Ожидаемая ставка рискованного проекта*	12%				
4	*Стандартное отклонение для рискованного проекта*	13%				
5	*Ставка доходности безрискового проекта*	6%				
6	*Процентная ставка займа*	8%				
7	*Оптимальная ставка доходности*	7%				
8	*Оптимальное стандартное отклонение*	3%				
9	*Часть инвестиций в рискованный проект*	22%				
10	*Коэффициент k*	1				
11	*Коэффициент p*	1				
12	*Минимизируемая функция*	1,22E-05				
13	*Ожидаемая доходность*	7%				
14	*Ожидаемое отклонение*	3%				
15	*Премия за риск*	1%				
16						

Рис. 2.7. Этот результат уже был получен ранее для более простой модели

Данная усложненная модель имеет одну особенность, которую хочется отметить. Дело в том, что в принципе можно так подобрать оптимальные значения ставки доходности и стандартного отклонения, что у модели будет существовать не одно, а два оптимальных решения. В зависимости от начального приближения для доли рискованных инвестиций можно получить либо одно, либо другое решение.

Это характерная особенность самой модели, и она никак не связана со спецификой приложения Excel. При желании можно проанализировать и такую ситуацию, но в данном случае это кажется не очень актуальным.

Формирование инвестиционного портфеля

На рынке представлено огромное количество всевозможных ценных бумаг, и большинство из них являются рискованными. Поэтому нет ничего удивительного в том, что на практике приходится иметь дело с инвестиционными портфелями, состоящими из рискованных активов нескольких типов. Далее рассмотрим ряд практически важных задач, имеющих отношение к проблеме распределения инвестиций среди ценных бумаг с разной степенью риска.



B12 =B11/КОРЕНЬ(C10*D10)

	A	B	C	D
1	*Взаимозависимость ставок доходности*			
2				
3		Вероятность	Доходность акций А	Доходность акций Б
4	Сценарий 1	0,25	12%	13%
5	Сценарий 2	0,3	10%	9%
6	Сценарий 3	0,4	9%	7%
7	Сценарий 4	0,05	5%	5%
8	Матожидание		9,85%	9,00%
9	Матожидание произведения	0,92%		
10	Дисперсия		0,03%	0,06%
11	Ковариация	0,04%		
12	Коэфициент корреляции	0,926663906		
13				

Рис. 2.8. Прямая корреляция между ставками доходности

В первую очередь, хочется отметить тот факт, что ставки доходности по ценным бумагам разных типов могут оказаться взаимозависимыми. Другими словами, между различными типами динамики доходности активов прослеживается некоторое соответствие, и в этом случае говорят о статистической корреляции между ставками доходности ценных бумаг. Численной характеристикой такой корреляции может служить ковариация ставок доходности, которая вычисляется как разность математического ожидания произведения соответствующих случайных величин и произведения их математических ожиданий. Однако более удобным с практической точки зрения является такая характеристика, как коэффициент корреляции, который, по сути своей, является ковариацией, нормированной на корень квадратный из произведения дисперсий соответствующих случайных величин. Удобство состоит в том, что значение коэффициента корреляции лежит в пределах от –1 до 1, а это существенно облегчает формальный анализ статистических или вероятностных зависимостей. Следует иметь в виду, что в принципе коэффициент корреляции может вычисляться как по прогнозируемым, так и по статистическим данным. Методика его расчета в каждом из этих случаев своя, однако радует то обстоятельство, что в Excel для каждого из этих случаев имеется специальная встроенная функция, которая позволяет вычислять соответствующие значения либо сразу, либо в несколько этапов. На рис 2.8 приведен пример документа с анализом взаимозависимости ставок доходности для различных ценных бумаг.

Прогнозируются четыре сценария развития ситуации, каждый из которых характеризуется определенной вероятностью реализации и набором значений ставок доходности по акциям двух типов А и Б. По этим данным вычисляется математическое ожидание (ожидаемое значение) для каждой из ставок доходности, а также математическое ожидание произведения этих ставок. Последняя характеристика понадобится при вычислении коэффициента корреляции. Формулы, вводимые в ячейки представленного документа, собраны в табл. 2.3.

Таблица 2.3. Назначение ячеек на рис. 2.8

Ячейка	Формула	Описание
C8	=СУММПРО-ИЗВ(B4:B7;C4:C7)	Вычисление математического ожидания ставки доходности акций типа А. Значением является сумма произведений возможного значения ставки на вероятность реализации этого значения
D8	=СУММПРО-ИЗВ(B4:B7;D4:D7)	Вычисление математического ожидания ставки доходности акций типа Б
B9	=СУММПРО-ИЗВ(B4:B7;C4:C7*D4:D7)	Вычисление математического ожидания произведения ставок доходности обоех типов акций. Значение вычисляется как сумма произведений значений ставок доходности акций каждого типа на вероятность реализации соответствующего сценария. Поскольку предварительно вычисляется массив произведений ставок доходности (команда C4:C7*D4:D7), то формула вводится как формула массива, то есть путем нажатия комбинации клавиш Ctrl+Shift+Enter
C10	=СУММПРО-ИЗВ(B4:B7;(C4:C7-C8)^2)	Дисперсия для ставки доходности акций А. Вычисляется как сумма произведений вероятностей реализации сценария на квадрат отклонения соответствующего значения ставки от ее математического ожидания. В качестве одного из аргументов указан массив квадратов отклонений значений ставки от математического ожидания — вычисляется как (C4:C7-C8)^2, поэтому формула вводится как формула массива

продолжение ⇨

Таблица 2.3 *(продолжение)*

Ячейка	Формула	Описание
D10	=СУММПРО-ИЗВ(B4:B7;(D4:D7-D8)^2)	Дисперсия для ставки доходности акций типа Б. Вычисляется аналогично предыдущему случаю. Формула массива
B11	=B9-C8*D8	Вычисление ковариации для ставок доходности акций. Определяется как разность математического ожидания произведения ставок доходности акций и произведения математических ожиданий этих величин
B12	=B11/КОРЕНЬ(C10*D10)	Вычисление коэффициента корреляции для ставок доходности акций. Определяется как отношение ковариации к корню квадратному из произведения дисперсий

Что касается непосредственно коэффициента корреляции, то в документе на рис. 2.8 он весьма близок к единице, что свидетельствует о высокой степени взаимозависимости между значениями ставок доходности акций. Рост ставок доходности по одной из акций А предполагает рост ставок доходности и по другой, снижение ставок доходности происходит также синхронно. На рис. 2.9 для акций Б изменены значения ставок доходности (они размещены в обратном порядке). В этом случае коэффициент корреляции отрицателен и близок к минус единице.

C10 fx {=СУММПРОИЗВ(B4:B7;(C4:C7-C8)^2)}

	A	B	C	D
1	***Взаимозависимость ставок доходности***			
2				
3		Вероятность	Доходность акций А	Доходность акций Б
4	Сценарий 1	0,25	12%	5%
5	Сценарий 2	0,3	10%	7%
6	Сценарий 3	0,4	9%	9%
7	Сценарий 4	0,05	5%	13%
8	Матожидание		9,85%	7,60%
9	Матожидание произведения	0,72%		
10	Дисперсия		0,03%	0,04%
11	Ковариация	-0,03%		
12	Коэфициент корреляции	-0,98524216		
13				

Рис. 2.9. Обратная корреляция между ставками доходности

Это означает, что между ставками доходности акций существует обратная корреляция: рост ставок доходности по одной из акций Б предполагает падение ставок доходности по другой и наоборот.

Выше коэффициент корреляции вычислялся на основе прогнозируемых данных. Другими словами, предполагалась известной вероятность реализации каждого из сценариев (равно как и возможные ставки доходности). В действительности рассчитывать степень коррелированности между ставками доходности по ценным бумагам приходится на основе статистических данных. В этом случае, как отмечалось, процедура вычисления коэффициента корреляции (равно как и прочих статистических характеристик) несколько иная. Кроме того, при вычислении статистических параметров можно с успехом использовать встроенные функции Excel, специально для этого предназначенные. Пример приведен на рис. 2.10.



B20 =КОРРЕЛ(B4:B15;C4:C15)

	A	B	C	D	E
1	***Вычисление коэффициента корреляции по статистическим данным***				
2					
3	**Месяц**	**Ставка доходности акций А**	**Ставка доходности акций Б**		
4	***Январь***	5%	6%		
5	***Февраль***	7%	7%		
6	***Март***	3%	4%		
7	***Апрель***	2%	3%		
8	***Май***	5%	2%		
9	***Июнь***	6%	4%		
10	***Июль***	3%	5%		
11	***Август***	6%	6%		
12	***Сентябрь***	8%	2%		
13	***Октябрь***	4%	1%		
14	***Ноябрь***	1%	3%		
15	***Декабрь***	7%	7%		
16					
17	***Среднее значение***	4,75%	4,17%		
18	***Дисперсия***	0,04%	0,04%		
19	***Ковариация***	0,0137500%	0,0137500%		
20	***Коэффициент корреляции***	0,337785607	0,337785607		
21					

Рис. 2.10. Вычисление коэффициента корреляции на основе статистических данных

В документе приведено два набора статистических данных для значений ставок доходности по акциям А и Б на протяжении календарного года. На основе этих данных вычисляются среднее значение ставки доходности, дисперсии, а также ковариация и коэффициент корреляции. Причем последние два вычисляются разными способами: напрямую с помощью специальных встроенных функций Excel и через стандартные математические и статистические функции. Данные относительно использованных формул приведены в табл. 2.4.

Таблица 2.4. Назначение ячеек на рис. 2.10

Ячейка	Формула	Описание
B17	=СРЗНАЧ(B4:B15)	Вычисление средней ставки доходности за период для акций А

продолжение ⇫

Таблица 2.4 *(продолжение)*

Ячейка	Формула	Описание
B18	=ДИСПР(B4:B15)	Вычисление дисперсии (по генеральной совокупности) для ставки доходности акций А
B19	=КОВАР(B4:B15;C4:C15)	Вычисление ковариации с помощью встроенной функции Excel
B20	=КОРРЕЛ(B4:B15;C4:C15)	Вычисление коэффициента корреляции с помощью встроенной функции Excel
C17	=СРЗНАЧ(C4:C15)	Вычисление средней ставки доходности за период для акций Б
C18	=ДИСПР(C4:C15)	Вычисление дисперсии (по генеральной совокупности) для ставки доходности акций Б
C19	=СРЗНАЧ(B4:B15*C4:C15)-B17*C17	Вычисление ковариации альтернативным способом. Формула вводится как формула массива
C20	=C19/КОРЕНЬ(B18*C18)	Коэффициент корреляции вычисляется как ковариация, деленная на корень квадратный из произведения дисперсий. Дисперсии вычисляются как для генеральной совокупности

В данном случае речь идет о выборочных характеристиках (то есть о характеристиках, которые вычисляются на основе данных выборки). Обычно в этом случае вычисляется выборочная дисперсия (в Excel для этих целей используют функцию `ДИСП()`). Однако, поскольку при вычислении коэффициента корреляции используют дисперсию по генеральной совокупности (это, как известно, смещенная оценка), в документе использована функция `ДИСПР()` (вычисление дисперсии для генеральной совокупности).

Что касается значения коэффициента корреляции, то он достаточно мал, поэтому можно предположить, что в рассмотренном примере ставки доходности по акциям изменяются независимо друг от друга. Но не следует забывать, что из равенства нулю коэффициента корреляции независимость случайных величин не следует.

Приведенный выше анализ сам по себе мало полезен. Обычно подобные задачи решаются в рамках более масштабных исследований по формированию инвестиционных портфелей. Далее рассмотрим еще одну упрощенную и в некотором смысле вспомогательную задачу. Требуется распределить инвестируемые средства между двумя рискованными активами. Для определенности будем полагать, что первый актив характеризуется ожидаемой доходностью ξ_1 и дисперсией σ_1^2, а второй — ожидаемой ставкой доходности ξ_2 и дисперсией σ_2^2. Предположим также, что в первый фонд вкладывают часть средств, равную

x (тогда во второй фонд вкладывают часть средств, равную $1 - x$). Можно показать, что в этом случае ожидаемая ставка доходности такого портфеля составляет $\xi = x\xi_1 + (1-x)\xi_2$, а дисперсия $\sigma^2 = x^2\sigma_1^2 + (1-x)^2\sigma_2^2 + 2x(1-x)\eta_{12}$, где через η_{12} обозначена ковариация для ставок доходности каждого из активов. Принимая во внимание, что коэффициент корреляции $\rho_{12} = \eta_{12}/(\sigma_1\sigma_2)$, можем также записать $\sigma^2 = x^2\sigma_1^2 + (1-x)^2\sigma_2^2 + 2x(1-x)\sigma_1\sigma_2\rho_{12}$. Из этих соотношений следует ряд достаточно интересных результатов. Во-первых, ожидаемая ставка доходности является линейной функцией параметра x, поэтому максимальной она будет либо при $x = 0$, либо при $x = 1$, в зависимости от значений ожидаемых ставок доходности по каждому из активов. Причем оптимизировать портфель путем выбора максимальной ожидаемой ставки доходности нет смысла. Во-вторых, дисперсия является квадратичной функцией параметра x. Минимальное значение дисперсии для портфеля достигается при $x = \dfrac{\sigma_2(\sigma_2 - \sigma_1\rho_{12})}{\sigma_1^2 + \sigma_2^2 - 2\sigma_1\sigma_2\rho_{12}}$. Если ставки доходности по каждому из активов изменяются независимо друг от друга, то коэффициент корреляции равен нулю и выражение для дисперсии портфеля примет вид $\sigma^2 = x^2\sigma_1^2 + (1-x)^2\sigma_2^2$, а минимальной дисперсия будет при $x = \dfrac{\sigma_2^2}{\sigma_1^2 + \sigma_2^2}$. В-третьих, поскольку и ожидаемая ставка доходности портфеля, и дисперсия являются функциями параметра x, можем рассматривать функциональную зависимость между ожидаемой ставкой доходности портфеля и дисперсией, причем зависимость задана в параметрическом виде, а параметром параметрической зависимости является переменная x. Именно с построения такой зависимости и начнем рассмотрение задачи по формированию портфеля. На рис. 2.11 представлены данные, на основе которых строится зависимость.

Документ условно состоит из трех частей. Ячейки B4 и C4 содержат значения ожидаемых ставок доходности по акциям А и Б, а в ячейках B5 и C5 представлены стандартные отклонения (корень квадратный из дисперсии) для ставок доходности акций. Коэффициент корреляции ставок доходности акций вводится в ячейку B6. Фактические данные, по которым строится кривая зависимости между ожидаемой ставкой доходности портфеля и стандартным отклонением (кривая инвестиционных возможностей), представлены в ячейках B9:C29. Ячейки A9:A29 содержат последовательность значений параметра x в диапазоне от 0 до 1 с шагом 0,05. Для каждого из этих значений вычисляются ожидаемая ставка доходности портфеля и стандартное отклонение. Эти данные представлены в диапазонах ячеек B9:B29 и C9:C29 соответственно. Заполняются ячейки следующим образом. В ячейку B9 вводится формула =A9*B4+(1-A9)*C4, после чего она копируется в прочие ячейки диапазона B9:B29. Диапазон ячеек C9:C29 заполняется копированием формулы =КОРЕНЬ(A9^2*B5^2+(1-A9)^2*C5^2+2*A9*(1-A9)*B5*C5*B6) из ячейки C9.

C9 | f_x =КОРЕНЬ(A9^2*B5^2+(1-A9)^2*C5^2+2*A9*(1-A9)*B5*C5*B6)

	A	B	C
1	*Кривая инвестиционных возможностей*		
2			
3		**Фонд акций А**	**Фонд акций Б**
4	**Ожидаемая ставка доходности**	12%	10%
5	**Стандартное отклонение**	5%	3%
6	**Коэффициент корреляции**	0	
7			
8	**Доля инвестиций x**	**Ставка доходности**	**Стандартное отклонение**
9	**0**	10,0%	3,00%
10	**0,05**	10,1%	2,86%
11	**0,1**	10,2%	2,75%
12	**0,15**	10,3%	2,66%
13	**0,2**	10,4%	2,60%
14	**0,25**	10,5%	2,57%
15	**0,3**	10,6%	2,58%
16	**0,35**	10,7%	2,62%
17	**0,4**	10,8%	2,69%
18	**0,45**	10,9%	2,79%
19	**0,5**	11,0%	2,92%
20	**0,55**	11,1%	3,06%
21	**0,6**	11,2%	3,23%
22	**0,65**	11,3%	3,42%
23	**0,7**	11,4%	3,61%
24	**0,75**	11,5%	3,82%
25	**0,8**	11,6%	4,04%
26	**0,85**	11,7%	4,27%
27	**0,9**	11,8%	4,51%
28	**0,95**	11,9%	4,75%
29	**1**	12,0%	5,00%
30			



Рис. 2.11. Кривая инвестиционных возможностей

При построении кривой инвестиционных возможностей в качестве типа диаграммы рекомендуется выбрать точечную диаграмму с соединением точек сглаженными кривыми.



Рис. 2.12. Кривая инвестиционных возможностей для значения коэффициента корреляции, равного 0,5